// File: /home/pangwa/walle-eth/contracts/WalleBot.sol



pragma solidity ^0.5.0;

import "@openzeppelin/contracts/token/ERC20/IERC20.sol";

// interface for legacy ERC20 tokens e.g. USDT
interface IERC20Legacy {
  function transfer(address _to, uint _value) external;
  function balanceOf(address who) external view returns (uint);
}

contract WalleBot {
  event TransferSuccess(IERC20 token);
  event OwnerAdded(address owner);
  event OwnerRemoved(address owner);

  address[4] _owners;

	constructor() public {
    _owners[0] = msg.sender;
	}

  modifier onlyOwner() {
      require(isOwner(), "Ownable: caller is not the owner");
      _;
  }

  function isOwner() public view returns (bool) {
      return isOwnerForAddress(msg.sender);
  }

  function isOwnerForAddress(address addr) public view returns (bool) {
      if (addr == address(0)) {
        return false;
      }
      for (uint i = 0; i < _owners.length; i++) {
        if (addr == _owners[i]) {
          return true;
        }
      }
      return false;
  }

  function owners() public view returns (address[4] memory) {
    return _owners;
  }

  function setOwner(uint index, address addr) public onlyOwner {
    require(index < 4, "invalid index!");
    address oldAddr = _owners[index];
    _owners[index] = addr;
    if (oldAddr != address(0)) {
      emit OwnerRemoved(oldAddr);
    }
    emit OwnerAdded(addr);
  }


  function removeOwner(uint index) public onlyOwner {
    require(index < 4, "invalid index!");
    address addr = _owners[index];
    _owners[index] = address(0);
    emit OwnerRemoved(addr);
  }

	function batchSend(IERC20 token, address[] memory receiver, uint[] memory amounts) public onlyOwner returns(bool sufficient) {
    require(receiver.length == amounts.length, "WalleBot: address should match amounts");
    for (uint i = 0; i < receiver.length; i++) {
      require(token.transfer(receiver[i], amounts[i]), "WalleBot: transfer failed");
    }
    emit TransferSuccess(token);
		return true;
	}

	function getBalance(IERC20 token) public view returns(uint) {
		return token.balanceOf(address(this));
	}

  function withdraw(IERC20 token) public onlyOwner returns (bool) {
    token.transfer(msg.sender, token.balanceOf(address(this)));
    return true;
  }

  // allow withdraw legacy erc 20 tokens
  function withdrawLegacy(IERC20Legacy token) public onlyOwner returns (bool) {
    token.transfer(msg.sender, token.balanceOf(address(this)));
    return true;
  }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



pragma solidity ^0.5.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP. Does not include
 * the optional functions; to access them see {ERC20Detailed}.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}